

## Postestimation commands

The following postestimation commands are available after `hetregress`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of parameters
* <code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC, respectively)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
* <code>estat (svy)</code>	postestimation statistics for survey data
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
† <code>forecast</code>	dynamic forecasts and simulations
† <code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
<code>linktest</code>	link test for model specification
*† <code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>predict</code>	linear predictions and their SEs, etc.
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of parameters
* <code>suest</code>	seemingly unrelated estimation
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

\*`estat ic`, `estat (svy)`, `lrtest`, and `suest` are not appropriate after `hetregress`, `twostep`.

†`forecast`, `hausman`, and `lrtest` are not appropriate with `svy` estimation results.

# predict

## Description for predict

`predict` creates a new variable containing predictions such as linear predictions, standard errors, and standard deviations.

## Menu for predict

Statistics > Postestimation

## Syntax for predict

After ML or two-step

```
predict [type] newvar [if] [in] [ , statistic]
```

After ML

```
predict [type] stub* [if] [in], scores
```

<i>statistic</i>	Description
Main	
<code>xb</code>	linear prediction; the default
<code>stdp</code>	standard error of the linear prediction
<code>sigma</code>	standard deviation of the error term

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

## Options for predict

### Main

`xb`, the default, calculates the linear prediction.

`stdp` calculates the standard errors of the linear prediction.

`sigma` calculates the standard deviations of the error term.

`scores` calculates equation-level score variables.

The first new variable will contain the derivative of the log likelihood with respect to the regression equation,  $\partial \ln L / \partial (\mathbf{x}_i \boldsymbol{\beta})$ .

The second new variable will contain the derivative of the log likelihood with respect to the scale equation (`lnsigma2`),  $\partial \ln L / \partial (\mathbf{z}_i \boldsymbol{\alpha})$ .

# margins

## Description for margins

`margins` estimates margins of response for linear predictions and of standard deviations.

## Menu for margins

Statistics > Postestimation

## Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
<code>xb</code>	linear prediction; the default
<code>stdp</code>	not allowed with <code>margins</code>
<code>sigma</code>	standard deviation of the error term

Statistics not allowed with `margins` are functions of stochastic quantities other than  $e(b)$ .

For the full syntax, see [R] [margins](#).

## Remarks and examples

Once you have fit a heteroskedastic regression model, you can use the `predict` command to obtain the predicted values both for the estimation sample and for other samples; see [U] [20 Estimation and postestimation commands](#) and [R] [predict](#). `predict` without arguments calculates the linear prediction from the fitted model  $\mathbf{x}_i\mathbf{b}$ , where  $\mathbf{x}_i$  are the independent variables in the  $j$ th observation and  $\mathbf{b}$  is the estimated parameter vector for the mean model. With the `stdp` option, `predict` calculates the standard error of the linear prediction. With the `sigma` option, `predict` calculates the predicted standard deviations of the error term,  $\hat{\sigma}_j = \exp(0.5 \times \mathbf{z}_i\mathbf{a})$ , where  $\mathbf{g}$  is the estimated parameter vector for the variance model.

▷ Example 1: Predicting heteroskedastic standard deviation

We can use `predict` to compute the predicted values of the standard deviations for female and male faculty based on the model from [example 2](#) in [R] [hetregress](#).

```
. use https://www.stata-press.com/data/r19/salary
(DeMaris (2004) - Faculty salaries)

. hetregress salary i.female##(c.priorexp c.yrrank c.yrbg c.salfac),
> het(i.female) twostep
(output omitted)

. predict sigma, sigma
. tabulate female, summarize(sigma)
```

1 = female; 0 = male	Summary of Heteroskedastic standard deviation		
	Mean	Std. dev.	Freq.
0	7741.8481	0	511
1	5828.6973	0	214
Total	7177.1388	873.22924	725

The predicted standard deviation for male faculty is  $7742/5829 \approx 1.3$  times the size for female faculty. We could have obtained the same results using `margins` with the `predict(sigma)` option.



▷ Example 2: Marginal means

We can use `margins` to compute the adjusted mean salary for male and female faculty when other factors are fixed at their means:

```
. margins female, atmeans
Adjusted predictions                               Number of obs = 725
Model VCE: Conventional
Expression: Linear prediction, predict()
At: priorexp = 2.89931 (mean)
    yrrank   = 7.397241 (mean)
    yrbg     = 12.52966 (mean)
    salfac   = .9399862 (mean)
```

	Delta-method				
	Margin	std. err.	z	P> z	[95% conf. interval]
female					
0	49036.29	420.3711	116.65	0.000	48212.37 49860.2
1	43822.74	569.9368	76.89	0.000	42705.69 44939.8

If everyone in the population were male faculty while holding all other factors at their mean values, the average salary would be \$49,036. If, instead, everyone were female faculty, the average salary would be \$43,823.



Also see

[R] [hetregress](#) — Heteroskedastic linear regression

## [U] 20 Estimation and postestimation commands

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

